

## 9/4.4.10

# Conversion de coordonnées polaires/cartésiennes

---

Les lycéens et étudiants sont souvent confrontés à des problèmes de conversion de coordonnées polaires en coordonnées rectangulaires et inversement. Les deux RSX présentées dans ce chapitre réalisent et facilitent à l'extrême ces opérations.

### COMMENT UTILISER LA RSX I POL

La RSX IPOL convertit des coordonnées rectangulaires en coordonnées polaires. Si vous désirez l'utiliser sous la forme Assembleur, saisissez le programme qui suit.

```

                ORG  9000H
                LOAD 9000H

;-----
; RSX POL
; Format : |POL,X,Y
; Entree : X=Abscisse
;         Y=Ordonnee
; Sortie : X=Rayon R
;         Y=Angle Teta
;-----
;
;
;-----
; Declaration des constantes
; et des variables du programme
;-----
;
RACFLO:      EQU  0BD9AH      ;Rac flottant
ADDFLO:      EQU  0BD79H      ;Addition flot
MULFLO:      EQU  0BD82H      ;Multipl flot
ENTFLO:      EQU  0BD61H      ;Entier->Flottant
DIVFLO:      EQU  0BD85H      ;Division flot
ATNFLO:      EQU  0BDB2H      ;Arc tangente
SIGFLO:      EQU  0BD91H      ;Signe d'1 flot
LOGEXT:      EQU  0BCD1H      ;KL LOG EXT
UNIT:        EQU  0BD94H      ;Unite de calcul
;
BUF:         DS    4          ;ZONE RAM POUR LOG EXT
9004 0990    PTRTAB:        DW  TABLE      ;Pointeur TABLE
9006 C32F90          JP    POL            ;Conversion

```

```

9009 504F  TABLE:      DB  "PO"
900B CC           DB  "L"+80H
900C 00           DB  0                ;Fin de table
;
SAVP1:          DS  2                ;Adresse 1er param
SAVP2:          DS  2                ;Adresse 2eme param
Z1:             DS  5                ;Zone reel 1
Z2:             DS  5                ;Zone reel 2
Z3:             DS  5                ;Zone reel 3
Z4:             DS  5                ;Zone reel 4
;
;-----
; Definition de la RSX
;-----
;
DEFRSX:         EQU  $                ;Point d'entree
9025 010490     LD  BC,PTRTAB        ;Ptr table definition
9028 210090     LD  HL,BUF           ;Buffer pour LOG EXT
902B CDD1BC     CALL LOGEXT          ;Definition de la RSX
902E C9         RET
;
;
;-----
; Traitement de POL
;-----
;
POL:            EQU  $                ;Point d'entree
;
;-----
; Lecture des donnees passees

```

```

;-----
;
902F DD6601      LD   H, (IX+1)
9032 DD6E00      LD   L, (IX+0)
9035 220F90      LD   (SAVP2),HL      ;Sauv 2eme argument
9038 CDD390      CALL ZONE2          ;Memo 2eme argument
903B DD6603      LD   H, (IX+3)
903E DD6E02      LD   L, (IX+2)
9041 220D90      LD   (SAVP1),HL      ;Sauv 1er argument
9044 CDCA90      CALL ZONE1          ;Memo 1er argument
;
;-----
; Calcul du rayon R
;-----
;
9047 211190      LD   HL, Z1
904A CDE590      CALL ZONE4          ;Copie de Z1 -> Z4
904D 212090      LD   HL, Z4
9050 112090      LD   DE, Z4
9053 CD82BD      CALL MULFLO        ;Z4 <- X^2
9056 211690      LD   HL, Z2
9059 CDDC90      CALL ZONE3          ;Copie de Z1 -> Z3
905C 211B90      LD   HL, Z3
905F 111B90      LD   DE, Z3
9062 CD82BD      CALL MULFLO        ;Z3 <- Y^2
9065 211B90      LD   HL, Z3
9068 112090      LD   DE, Z4
906B CD79BD      CALL ADDFLO        ;Z3 <- X^2+Y^2
906E 211B90      LD   HL, Z3
9071 CD9ABD      CALL RACFLO        ;Z3 <- Rac(X^2+Y^2)

```

```

9074 211B90          LD  HL,Z3
9077 ED5B0D90       LD  DE,(SAVP1)
907B CDEE90         CALL FLODEHL          ;Rayon dans P1
                    ;
                    ;- - - - -
                    ; Calcul de l'angle Teta
                    ;- - - - -
                    ;

907E 3EFF           LD  A,OFFH
9080 CD94BD        CALL UNIT          ;Calculs en degres
9083 211690       LD  HL,Z2
9086 111190       LD  DE,Z1
9089 CD85BD        CALL DIVFLO
908C 211190       LD  HL,Z1
908F CD91BD        CALL SIGFLO
9092 FEFF         CP   255
9094 CAA690       JP   Z,XNEG          ;X negatif
9097 211690       LD  HL,Z2
909A CDB2BD        CALL ATNFLO
909D ED5B0F90     LD  DE,(SAVP2)
90A1 CDEE90       CALL FLODEHL          ;Angle dans P2
90A4 1823         JR   FIN
                    ;
XNEG:             EQU  $
90A6 CDB2BD        CALL ATNFLO
90A9 AF           XOR  A
90AA 21B400       LD  HL,180
90AD 111B90       LD  DE,Z3
90B0 CD61BD        CALL ENTFLO
90B3 211690       LD  HL,Z2

```

```
90B6 CDB2BD          CALL ATNFLO
90B9 211690          LD   HL,Z2
90BC 111B90          LD   DE,Z3
90BF CD79BD          CALL ADDFLO
90C2 ED5B0F90        LD   DE,(SAVP2)
90C6 CDEE90          CALL FLODEHL          ;Angle dans P2
;
FIN:                 EQU  $
90C9 C9              RET
;
;-----
; Zone des sous-programmes
;-----
;
;-----
; Transfert des BC octets pointes
; par HL dans Z1, Z4 ou Z5
;-----
;
ZONE1:               EQU  $
90CA 111190          LD   DE,Z1
90CD 010500          LD   BC,5
90D0 EDB0            LDIR
90D2 C9              RET
;
ZONE2:               EQU  $
90D3 111690          LD   DE,Z2
90D6 010500          LD   BC,5
90D9 EDB0            LDIR
90DB C9              RET
```

```

;
ZONE3:      EQU  $
90DC 111B90      LD  DE,Z3
90DF 010500      LD  BC,5
90E2 EDB0        LDIR
90E4 C9          RET

;
ZONE4:      EQU  $
90E5 112090      LD  DE,Z4
90E8 010500      LD  BC,5
90EB EDB0        LDIR
90ED C9          RET

;
;-----
; Transfert flottant de (HL)
; dans (DE)
;-----
;
FLODEHL:    EQU  $
90EE 010500      LD  BC,5
90F1 EDB0        LDIR
90F3 C9          RET

END

```

Activez la RSX en tapant :

**CALL &9025**

Il vous suffit ensuite de lui passer les valeurs **X** et **Y** à convertir. En sortie de RSX, ces variables contiendront les valeurs **R** et **Téta**.

*Exemple :*

**A=4 : B=3 : I POL, @A, @B : PRINT A,B**

Si vous préférez utiliser un chargeur Basic, saisissez le listing suivant :

```

1000 REM -----
1010 REM TRANSFORMATION RECTANGULAIRE->POLAIRE
1020 REM -----
1030 '
1040 ' - - - - -
1050 ' Lecture des donnees et memorisation
1060 ' - - - - -
1070 '
1080 FOR i=&9000 TO &90F3
1090   READ a$
1100   a=VAL("&"+a$)
1110   POKE i,a
1120 NEXT i
1130 '
1140 '
1150 ' - - - - -
1160 ' Definition de la RSX POL
1170 ' - - - - -
1180 '
1190 CALL &9025
1200 '
1210 ' - - - - -
1220 ' Activation de la RSX POL
1230 ' - - - - -
1240 '
1250 a=4 : b=3
1260 |POL,@a,@b
1270 PRINT "R=";a;" Teta=";b
1280 END
1290 '
1300 REM -----
1310 REM Donnees du programme Assembleur
1320 REM -----
1330 '
1340 DATA 0,0,0,0,9,90,C3,2F,90,50,4F,CC,0,0,0,0
1350 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
1360 DATA 0,0,0,0,0,1,4,90,21,0,90,CD,D1,BC,C9,DD
1370 DATA 66,1,DD,6E,0,22,F,90,CD,D3,90,DD,66,3,DD,6E
1380 DATA 2,22,D,90,CD,CA,90,21,11,90,CD,E5,90,21,20,90
1390 DATA 11,20,90,CD,82,BD,21,16,90,CD,DC,90,21,1B,90,11
1400 DATA 1B,90,CD,82,BD,21,1B,90,11,20,90,CD,79,BD,21,1B
1410 DATA 90,CD,9A,BD,21,1B,90,ED,5B,D,90,CD,EE,90,3E,FF
1420 DATA CD,94,BD,21,16,90,11,11,90,CD,85,BD,21,11,90,CD
1430 DATA 91,BD,FE,FF,CA,A6,90,21,16,90,CD,B2,BD,ED,5B,F
1440 DATA 90,CD,EE,90,18,23,CD,B2,BD,AF,21,B4,0,11,1B,90
1450 DATA CD,61,BD,21,16,90,CD,B2,BD,21,16,90,11,1B,90,CD
1460 DATA 79,BD,ED,5B,F,90,CD,EE,90,C9,11,11,90,1,5,0
1470 DATA ED,B0,C9,11,16,90,1,5,0,ED,B0,C9,11,1B,90,1
1480 DATA 5,0,ED,B0,C9,11,20,90,1,5,0,ED,B0,C9,1,5
1490 DATA 0,ED,B0,C9,0,0,0,0,0,0,0,0,0,0,0,0,0

```




Vérifiez que les données hexadécimales ont été bien entrées à l'aide des données de checksum suivantes :

**89 0 4B 3B C3 B0 89 F5 3C AE 99 45 EF 4C A3 68**

Les lignes 1250 à 1270 activent la RSX et affichent les valeurs retournées.

### LA RSX EN DÉTAIL

La logique de la RSX apparaît dans l'ordinogramme suivant :



Les premières lignes du programme sont les références des macros appelées, et les zones utilisées pour stocker les adresses et nombres réels manipulés.

Les lignes suivantes définissent la RSX. Ces instructions sont maintenant classiques. Nous n'y reviendrons pas.

L'adresse de traitement de la RSX est **POL**. Les premières instructions situées après cette adresse mémorisent :

- les données passées en paramètres dans les zones réelles **Z1** et **Z2** ;
- l'adresse des données passées en paramètres dans les variables **SAVP1** et **SAVP2**.

Le bloc d'instructions suivant calcule le rayon **R** avec la formule suivante :

$$R = \text{RACINE}(X^2+Y^2)$$

Les macros utilisées sont :

- **MULFLO** pour calculer le carré ;
- **ADDFLO** pour calculer la somme ;
- **RACFLO** pour calculer la racine carrée.

Le bloc d'instructions suivant calcule l'angle **Téta** avec la formule suivante :

$$\text{Téta} = \text{Arc Tangente}(Y/X)$$

Les calculs sont effectués en degrés :

```
LD    A,0FFH
CALL UNIT
```

En fonction du signe de **X** et de **Y**, la formule doit être réajustée comme suit :

Si $X > 0$ et $Y > 0$ ,	$\text{Téta} = \text{ATN}(Y/X)$
Si $X < 0$ et $Y > 0$ ,	$\text{Téta} = \text{ATN}(Y/X) + 180^\circ$
Si $X < 0$ et $Y < 0$ ,	$\text{Téta} = \text{ATN}(Y/X) + 180^\circ$
Si $X > 0$ et $Y < 0$ ,	$\text{Téta} = \text{ATN}(Y/X)$

Ce qui se résume en :

**Si  $X < 0$ , ajouter 180 degrés à l'arc tangente.**

Les dernières lignes du programme contiennent divers sous-programmes de transferts numériques réels déjà étudiés précédemment. Nous n'y reviendrons pas.

**COMMENT UTILISER LA RSX I RECT**

La RSX IRECT convertit des coordonnées polaires en coordonnées rectangulaires. Si vous désirez l'utiliser sous la forme Assembleur, saisissez le programme suivant :

```

1          ORG  9000H
2
3          LOAD 9000H
4
5          ;-----
6
7          ; RSX RECT
8
9          ; Format : |RECT,R,T
10
11         ; Entree : R=Rayon
12
13         ;          T=Angle Teta
14
15         ; Sortie : X=Abscisse
16
17         ;          Y=Ordonnee
18
19         ;-----
20
21         ;
22
23         ;
24
25         ;-----
26
27         ; Declaration des constantes
28
29         ; et des variables du programme
30
31         ;-----
32
33         ;
34
35         MULFLO:  EQU  0BD82H          ;Multipl flot
36
37         COSFLO:  EQU  0BDACH          ;Cosinus flot
38
39         SINFLO:  EQU  0BDA9H          ;Sinus flot
40
41         LOGEXT:  EQU  0BCD1H          ;KL LOG EXT
42
43         UNIT:    EQU  0BD94H          ;Unite de calcul
44
45         ;
46
47         BUF:     DS    4              ;ZONE RAM POUR LOG EXT
48
49 9004 0990      PTRTAB:  DW  TABLE          ;Pointeur TABLE
50
51 9006 C32B90    JP  RECT              ;Conversion
52

```

```

53 9009 524543  TABLE:      DB   "REC"
54
55 900C D4      DB   "T"+80H
56
57 900D 00      DB   0           ;Fin de table
58
59              ;
60
61 SAVP1:       DS   2           ;Adresse 1er param
62
63 SAVP2:       DS   2           ;Adresse 2eme param
64
65 Z1:         DS   5           ;Zone reel 1
66
67 Z2:         DS   5           ;Zone reel 2
68
69 Z3:         DS   5           ;Zone reel 3
70
71              ;
72
73              ;-----
74
75              ; Definition de la RSX
76
77              ;-----
78
79              ;
80
81 DEFRSX:     EQU  $           ;Point d'entree
82
83 9021 010490   LD   BC,PTRTAB   ;Ptr table definition
84
85 9024 210090   LD   HL,BUF       ;Buffer pour LOG EXT
86
87 9027 CDD1BC   CALL LOGEXT     ;Definition de la RSX
88
89 902A C9       RET
90
91              ;
92
93              ;
94
95              ;-----
96
97              ; Traitement
98
99              ;-----
100
101              ;
102
103 RECT:       EQU  $           ;Point d'entree
104
105              ;
106
107              ;-----
108
109              ; Lecture des donnees passees
110

```

```

111          ; - - - - -
112
113          ;
114
115 902B DD6601          LD   H, (IX+1)
116
117 902E DD6E00          LD   L, (IX+0)
118
119 9031 221090          LD   (SAVP2),HL          ;Sauv 2eme argument
120
121 9034 CD8A90          CALL ZONE2          ;Memo 2eme argument
122
123 9037 DD6603          LD   H, (IX+3)
124
125 903A DD6E02          LD   L, (IX+2)
126
127 903D 220E90          LD   (SAVP1),HL          ;Sauv 1er argument
128
129 9040 CD8190          CALL ZONE1          ;Memo 1er argument
130
131          ;
132
133          ; - - - - -
134
135          ; Calcul de l'abscisse X
136
137          ; - - - - -
138
139          ;
140
141 9043 3EFF            LD   A,0FFH
142
143 9045 CD94BD          CALL UNIT          ;Calculs en degres
144
145          ;
146
147 9048 211790          LD   HL,Z2
148
149 904B CD9390          CALL ZONE3          ;Sauvegarde en zone 3
150
151 904E 211790          LD   HL,Z2
152
153 9051 CDACBD          CALL COSFLO
154
155 9054 211790          LD   HL,Z2
156
157 9057 111290          LD   DE,Z1
158
159 905A CD82BD          CALL MULFLO
160
161 905D 211790          LD   HL,Z2
162
163 9060 ED5B0E90        LD   DE, (SAVP1)
164
165 9064 CD9C90          CALL FLODEHL          ;Abscisse dans P1
166
167          ;
168
169          ; - - - - -

```

```

170
171           ; Calcul de l'ordonnee Y
172
173           ;-----
174
175           ;
176
177 9067 211C90           LD   HL,Z3
178
179 906A CDA9BD           CALL SINFLO
180
182
183 9070 111290           LD   DE,Z1
184
185 9073 CD82BD           CALL MULFLO
186
187 9076 211C90           LD   HL,Z3
188
189 9079 ED5B1090        LD   DE,(SAVP2)
190
191 907D CD9C90           CALL FLODEHL           ;Angle dans P2
192
193           FIN:        EQU  $
194
195 9080 C9               RET
196
197           ;
198
199           ;-----
200
201           ; Zone des sous-programmes
202
203           ;-----
204
205           ;
206
207           ;-----
208
209           ; Transfert des BC octets pointes
210
211           ; par HL dans Z1, Z4 ou Z5
212
213           ;-----
214
215           ;
216
217           ZONE1:      EQU  $
218
219 9081 111290           LD   DE,Z1
220
221 9084 010500           LD   BC,5
222
223 9087 EDB0             LDIR
224
225 9089 C9               RET
226
227           ;

```

```

228
229           ZONE2:      EQU  $
230
231 908A 111790           LD   DE,Z2
232
233 908D 010500           LD   BC,5
234
235 9090 EDB0             LDIR
236
237 9092 C9               RET
238
239           ;
240
241           ZONE3:      EQU  $
242
243 9093 111C90           LD   DE,Z3
244
245 9096 010500           LD   BC,5
246
247 9099 EDB0             LDIR
248
249 909B C9               RET
250
251           ;
252
253           ;-----
254           ; Transfert flottant de (HL)
255           ; dans (DE)
256           ;-----
257
258           ;
259
260           FLODEHL:    EQU  $
261
262
263 909C 010500           LD   BC,5
264
265 909F EDB0             LDIR
266
267 90A1 C9               RET
268
269
270           END
271

```

Activez la RSX en tapant :

**CALL &9021**

Il vous suffit ensuite de lui passer les valeurs **R** et **Téta** à convertir. En sortie de la RSX, ces variables contiendront les valeurs **X** et **Y**.

*Exemple :*

**R=5 : T=30 : I RECT, @R,@T : PRINT R,T**

Si vous préférez utiliser un chargeur Basic, saisissez le listing suivant :

```

1000 REM -----
1010 REM TRANSFORMATION POLAIRE->RECTANGULAIRE
1020 REM -----
1030 '
1040 ' - - - - -
1050 ' Lecture des donnees et memorisation
1060 ' - - - - -
1070 '
1080 FOR i=&9000 TO &90A1
1090   READ a$
1100   a=VAL("&" + a$)
1110   POKE i,a
1120 NEXT i
1130 '
1140 '
1150 ' - - - - -
1160 ' Definition de la RSX RECT
1170 ' - - - - -
1180 '
1190 CALL &9021
1200 '
1210 ' - - - - -
1220 ' Activation de la RSX RECT
1230 ' - - - - -
1240 '
1250 a=5 : b=30
1260 |RECT,@A,@B
1270 PRINT "X=";a;" Y=";b
1280 END
1290 '
1300 REM -----
1310 REM Donnees du programme Assembleur
1320 REM -----
1330 '
1340 DATA 0,0,0,0,9,90,C3,2B,90,52,45,43,D4,0,0,0
1350 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
1360 DATA 0,1,4,90,21,0,90,CD,D1,BC,C9,DD,66,1,DD,6E
1370 DATA 0,22,10,90,CD,8A,90,DD,66,3,DD,6E,2,22,E,90
1380 DATA CD,81,90,3E,FF,CD,94,BD,21,17,90,CD,93,90,21,17
1390 DATA 90,CD,AC,BD,21,17,90,11,12,90,CD,82,BD,21,17,90
1400 DATA ED,5B,E,90,CD,9C,90,21,1C,90,CD,A9,BD,21,1C,90
1410 DATA 11,12,90,CD,82,BD,21,1C,90,ED,5B,10,90,CD,9C,90
1420 DATA C9,11,12,90,1,5,0,ED,B0,C9,11,17,90,1,5,0
1430 DATA ED,B0,C9,11,1C,90,1,5,0,ED,B0,C9,1,5,0,ED
1440 DATA B0,C9,0,0,0,0,0,0,0,0,0,0,0,0,0,0

```




Vérifiez que les données hexadécimales ont été entrées à l'aide des données de checksum suivantes :

**C8 0 FE 2 31 1C B3 74 AA 88 7A**

Les lignes 1250 à 1270 activent la RSX et affichent les valeurs retournées.

### LA RSX EN DÉTAIL

La logique de la RSX apparaît dans l'ordinogramme suivant :



Cette RSX est bien plus simple que la précédente. Pour convertir des coordonnées polaires en coordonnées rectangulaires, il suffit en effet :

- de multiplier le rayon par le cosinus de l'angle pour obtenir l'abscisse ;
- de multiplier le rayon par le sinus de l'angle pour obtenir l'ordonnée.

Les premières lignes du programme sont les références des macros appelées, et les zones utilisées pour stocker les adresses et nombres réels manipulés.

Les lignes suivantes définissent la RSX.

L'adresse de traitement de la RSX est **RECT**. Les premières instructions situées après cette adresse mémorisent :

- les données passées en paramètres dans les zones réelles **Z1** et **Z2** ;
- l'adresse des données passées en paramètres dans les variables **SAVP1** et **SAVP2**.

Le bloc d'instructions suivant calcule l'abscisse **X** avec la formule suivante :

$$X = R * \text{COS}(\text{Téta})$$

Les macros utilisées sont :

- **UNIT** pour définir l'unité des calculs trigonométriques en degrés ;
- **COSFLO** pour calculer le cosinus de **Téta** ;
- **MULFLO** pour effectuer la multiplication de **R** par le cosinus de l'angle.

Le bloc d'instructions suivant calcule l'ordonnée selon le même procédé. Nous n'y reviendrons pas.